
Abstract
The back-and-forth method is a well-known technique to establish isomorphisms of countable
structures. In this proof pearl, we formalise this method abstractly in the framework of constructive
type theory, emphasising the computational interpretation of the constructed isomorphisms. As
prominent instances, we then deduce Cantor’s and Myhill’s isomorphism theorems on dense linear
orders and one-one interreducible sets, respectively. By exploiting the symmetry of the abstract
argument, our approach yields a particularly compact mechanisation of the method itself as well as
its two instantiations, all implemented using the Coq proof assistant. As adequate for a proof pearl,
we attempt to make the text and mechanisation accessible for a general mathematical audience.

1 The Informal Argument

We begin by explaining the back-and-forth method informally to provide some background
and intuition, using the example of Cantor’s isomorphism theorem on countable dense linear
orders. Actually established by Cantor himself via an alternative but related method [2], his
result is often considered the origin of the back-and-forth argument and textbook presentations
typically follow this particular strategy (Hausdorff’s “Grundzüge der Mengenlehre” [8] may be
the earliest example).1 An important consequence of Cantor’s isomorphism theorem is that
the first-order theory of unbounded dense linear orders is complete and therefore decidable,
which is the reason why the back-and-forth method was brought to fruition with a variety of
similar results in model theory (see for instance [13], especially Chapter 6). While we do not
include more of these model-theoretic constructions in this paper, as a second instance of an
explicitly computational flavour we will consider Myhill’s isomorphism theorem [12], stating
that one-one interreducible sets are recursively isomorphic, at a later point.

So to get started on Cantor’s isomorphism theorem, recall that (strict) linear orders may
be characterised as irreflexive and transitive binary relations (X, <) satisfying trichotomy
(x < y ∨ x = y ∨ y < x). Such a linear order is called dense if whenever x < y, then there is
some z with x < z < y. These requirements alone do not yet fully determine the structure,
as two orderings might differ regarding their endpoints. This leaves four possible cases
depending on the existence of left and/or right endpoints, we focus on the unbounded version:

∀x. ∃yy′. y < x < y′

1 We refer the interested reader to [17] for further historic references as well as a comparison of Cantor’s
actual proof to the back-and-forth method.

Figure 1 Illustration of the three cases how a finite stage of the order-isomorphism of (X, <) and
(Y, <) can be extended. By construction, the stage f1 already contains matchings for x0 and y0.
For the construction of f2 we want to find a matching for the next element x1, which either lies in
between the domain of f1 or completely above or below it. In any case we find a matching partner
using density, upwards unboundedness, or downwards unboundedness of Y .

Now given two unbounded dense linear orders (X, <) and (Y, <) that are countable,
so assuming that X = {x0, x1, x2, . . . } and Y = {y0, y1, y2, . . . }, we want to establish a
bijection F : X → Y respecting the order, i.e. satisfying x < x′ iff F (x) < F (x′). This
order-isomorphism can be constructed in finite stages, meaning that we begin with the empty
partial isomorphism f0 and iteratively extend to f1, f2, . . . such that xn is in the domain
of fn+1. To make sure that in this process no element of Y is missed, we simultaneously
take care that yn is in the range of fn+1. Once the stages fn are defined, we simply set
F (xn) := fn+1(xn) and observe that the expected properties transport from the stages to F .
Borrowing common set-theoretic notation, the construction can be summarised as

f0 := ∅
fn+1 := {(xn, y), (x, yn)} ∪ fn

F :=
⋃

n∈N
fn

where the matching partners y and x for xn and yn, respectively, remain to be determined.
To this end, we begin with a forwards step treating xn. If fn(xn) is already defined, then

there is nothing to do. Otherwise, we need to find a suitable partner y not yet in the range
of fn which we can safely add as image of xn while preserving the orderings. We distinguish
three possible cases, which are also depicted in Figure 1:

If there are x and x′ in the domain of fn such that x < xn < x′, then (since fn is finite)
we can assume that x is the greatest such element while x′ is the smallest. By the density
of (Y, <) there is y with fn(x) < y < fn(x′) which is then a suitable match for x.
If the whole domain of fn is below xn, then using the unboundedness of (Y, <) we choose
y such that the whole range of fn is below y. This makes y the desired match for x.
In the remaining case where xn is below the domain of fn, we find a match y below the
range of fn again by unboundedness.

In any case we set f ′
n+1(xn) := y as an intermediate extension of fn and continue with the

backwards step treating yn, again only necessary if yn is not yet in the range of f ′
n+1. This

step is completely symmetric to the forward step, that is, we find a suitable match x for yn

by distinguishing the same three cases regarding the position of yn relative to the range of
f ′

n+1. Once a matching partner x is found, we extend f ′
n+1 by setting fn+1(x) := yn.

This concludes the construction of fn+1 and therefore of the order-isomorphism F in the
case of unbounded orders. We just remark that the constructions work analogously if (X, <)
and (Y, <) were to contain endpoints of the same characteristic.

In the next section, we continue by extracting the abstract idea of the above argument into
a general isomorphism theorem, which is then instantiated to Cantor’s result in Section 4. To
illustrate the generality, in Section 5 we provide a second instantiation to Myhill’s isomorphism
theorem, crucially relying on the computational perspective outlined in Section 3.

All results of these technical sections are formalised in constructive type theory and
mechanised in a single Coq file of about 500 lines of code.2 While (even constructive)
mechanisations of Cantor’s isomorphism theorem were already given Giese and Schönegge [6]
as well as Marzion [10], and while we closely follow Forster, Jahn, and Smolka [3] in the
instantiation to Myhill’s isomorphism theorem, we are not aware of an abstract formulation
in constructive type theory, let alone a mechanisation with a proof assistant, of the back-
and-forth method and its computational interpretation as such.

2 The Abstract Argument

Now switching to the formal framework of constructive type theory, we represent the two
sets X and Y subject to the back-and-forth argument by types in the universe T. Since
the method only applies to countable structures, we could fix X and Y to be the type N
of natural numbers. To explicitly include finite types, however, we just assume that X and
Y are retracts of N, so for instance X comes with an injection iX : X ↪→ N with explicitly
given surjective left-inverse rX : N ↠ X, where we just write xn for rX n.3 In order to
investigate the argument abstractly, we need to axiomatise three ingredients: the particular
structure of each instance, the corresponding notion of structure-preserving isomorphism,
and a procedure to extend a partial isomorphism by one step.

The Abstract Assumptions
We first abstract over the required structure relating both sides of the potential isomorphism by
an operation A : T×T → T. For Cantor’s isomorphism theorem, an element SX,Y of A(X, Y )
will independently equip both X and Y with a dense unbounded linear order. Myhill’s
isomorphism theorem, however, imposes shared structure via mutual one-one reductions,
making a binary operation necessary. To still enable the full usage of symmetry, we further
assume a function associating to every structured pair SX,Y its inverse S−1

X,Y in A(Y, X) with

(S−1
X,Y )−1 = SX,Y . (1)

The next ingredient is the abstract isomorphism property we want to establish. It suffices
to express this property locally, i.e. to relate pairs (x, x′) in X × X and (y, y′) in Y × Y ,
written (x, x′) ∼ (y, y′), relative to SX,Y . The idea is to let (x, x′) ∼ (y, y′) express that x

relates to x′ in the structure over X exactly like y relates to y′ in the structure over Y . In
particular, we require that the structural similarity extends to equality, formally

(x, x′) ∼ (y, y′) → (x = x′ ↔ y = y′) (2)

and that inverting the structure SX,Y to S−1
X,Y maintains the matching, formally

(x, x′) ∼ (y, y′) → (y, y′) ∼ (x, x′) (3)

2 The Coq file can be obtained from the URL listed as supplementary material and an HTML version is
accessible by simply clicking on any of the highlighted statements in this PDF.

3 Note that for simplicity we exclude empty types as retracts of N by the requirement that rX be total.
For our purpose, we deem this choice reasonable as empty structures are usually trivially isomorphic.

ITP 2022

Figure 2 Illustration of the abstract assumptions. Two types X and Y are related by a surrounding
structure SX,Y . If a partial isomorphism L connects an element x : X with y : Y , then for x′ the
step function yields an element µ x′ L relating to y like x′ relates to x, yielding (x, x′) ∼ (y, µ x′ L).

where the premise and conclusion are relative to SX,Y and S−1
X,Y , respectively. The desired

isomorphism can then be characterised as a function F : X → Y with inverse F −1 : Y → X

such that (x, x′) ∼ (F x, F x′) for all x and x′.
Before we get to the final third ingredient, we need to model the notion of (finite) partial

isomorphisms approximating F and F −1. A simple choice is to take lists4 L : L(X × Y ) of
pairs (x, y) : X × Y and call them partial isomorphisms, written SX,Y (L) to emphasise the
dependency on SX,Y , if (x, x′) ∼ (y, y′) whenever (x, y) ∈ L and (x′, y′) ∈ L. By flipping the
components of each pair in L, we can turn L into L−1 : L(Y × X) with the obvious property
that S−1

X,Y (L−1) whenever SX,Y (L). We denote by dom(L) and ran(L) the domain and range
of L, respectively, and write L[x] for the (first) value of x ∈ dom(L).

So finally concluding the abstract ingredients, we assume a polymorphic step function

µ : ∀XY. A(X, Y ) → X → L(X × Y ) → Y

finding a suitable partner y for x not yet included in L, meaning that if L is a partial
isomorphism and x ̸∈ dom(L), then (x, µ x L) :: L is a partial isomorphism:5

x ̸∈ dom(L) → SX,Y (L) → SX,Y ((x, µ x L) :: L) (4)

To sum up, we illustrate the abstract setup graphically in Figure 2 and provide an overview
of all assumptions as stated in Coq in Listing 1.

The Abstract Construction
As in the informal argument, the idea is to start with the empty partial isomorphism and
then to extend in step n + 1 by matching up both xn and yn with suitable partners (if not
yet accommodated), thereby ensuring that both structures are exhausted. So we first extend
µ to a polymorphic function µ′ adding a new match for x to L if no existing match is found:

µ′ : ∀XY. A(X, Y ) → X → L(X × Y ) → L(X × Y )

µ′ x L :=
{

(x, µ x L) :: L if x ̸∈ dom(L)
(x, L[x]) :: L if x ∈ dom(L)

4 Constructed from the empty list [] and the cons operation x :: L for x : X and L : L(X).
5 We treat the first three arguments of µ as implicit and will do so for all similar functions.

https://www.ps.uni-saarland.de/extras/back-and-forth/website/baf.backandforth.html#step
https://www.ps.uni-saarland.de/extras/back-and-forth/website/baf.backandforth.html#step

Listing 1 Abstract assumptions regarding the structure (structure), the corresponding notion of
isomorphism (iso), and the extension procedure (find) as formulated in Coq. Note that the former
two take their arguments in curried form as opposed to the cartesian representation on paper.

structure : Type -> Type -> Type
srev : forall X Y, structure X Y -> structure Y X
srev_invol : forall X Y (S : structure X Y), srev (srev S) = S

iso : forall X Y, structure X Y -> X -> X -> Y -> Y -> Prop
iso_eq : forall X Y (S : structure X Y) x x' y y', iso S x x' y y' -> x = x' <-> y = y'
iso_rev : forall X Y (S : structure X Y) x x' y y', iso S x x' y y' -> iso (srev S) y y' x x'

find : forall X, structure X Y -> X -> list (X * Y) -> Y
find_iso : forall X Y (S : structure X Y) L x, x ̸∈ dom L -> tiso S L -> tiso S ((x, find S x L) :: L)

Note that in the second case, where a match for x was already present in L, we could as
well just return L, but the choice to repeat the match slightly simplifies the (in any case
straightforward) argument for the desired property x ∈ dom(µ′ x L).

Now since µ′ is polymorphic and therefore usable in both directions, the iterative process at
the heart of the back-and-forth argument can be defined exploiting the structural symmetry:

L_ : ∀XY. A(X, Y ) → N → L(X × Y )

L0 := []
Ln+1 := (µ′ yn (µ′ xn Ln)−1)−1

So in the second case, we first extend Ln with a match for xn by the inner application of µ′

(regarding SX,Y ) and then flip the result to obtain a partial isomorphism in the opposite
direction. This is then extended with a match for yn by the outer application of µ′ (regarding
S−1

X,Y ), after which the result is flipped again to yield a partial isomorphism in the original
direction.

Since we have xn ∈ dom(Ln+1) and yn ∈ dom(Ln+1)−1 = ran(Ln+1) by the above remark
that x ∈ dom(µ′ x L) for all x, we can define the isomorphism F : X → Y and its inverse
F −1 : Y → X on a given structured pair SX,Y simply by

F xn := Ln+1[xn] and F −1 xn := L−1
n+1[yn]

where we implicitly use that for every x : X there is n : N with x = xn, analogously for y : Y .

The Abstract Verification
To conclude the abstract back-and-forth argument, we show that the function F : X → Y

is an isomorphism for SX,Y . We first formally state that flipping and extending with µ′

preserves partial isomorphisms.

▶ Lemma 1. If SX,Y (L) then S−1
X,Y (L−1) and SX,Y (µ′ x L).

Proof. Straightforward by Assumptions (3) and (4), respectively. ◀

Then in particular every Ln is a partial isomorphism.

▶ Lemma 2. SX,Y (Ln)

Proof. By induction on n, the case of L0 = [] is trivial. For Ln+1 we assume SX,Y (Ln)
as inductive hypothesis and apply both parts of Lemma 1 in both directions to obtain
(S−1

X,Y )−1(Ln+1), which is the same as SX,Y (L) by Assumption (1). ◀

ITP 2022

We can now prove the abstract isomorphism theorem with a simple argument.

▶ Theorem 3 (Isomorphism). F satisfies (x, x′) ∼ (F x, F x′) and is inverted by F −1.

Proof. For the first claim, let x = xn, x′ = xm, and w.l.o.g n ≤ m. Since then Ln+1 ⊆ Lm+1,
we have (x, F x) = (xn, Ln+1[x]) ∈ Lm+1 and (x′, F x′) = (xm, Lm+1[x]) ∈ Lm+1, from
which conclude (x, x′) ∼ (F x, F x′) since SX,Y (Lm+1) by Lemma 2.

For the second claim, note that for some n big enough, we have both (x, F x) ∈ Ln

(since x is eventually handled by some forwards step) and (F −1(F x), F x) ∈ Ln (since
F x is eventually handled by some backwards step). By Lemma 2 we therefore obtain
(x, F −1(F x)) ∼ (F x, F x), from which x = F −1(F x) follows by Assumption (2). Analog-
ously, we establish y = F (F −1 y) and thus conclude the claim that F −1 inverts F . ◀

3 The Computational Argument

Before we consider instantiations of our abstract back-and-forth argument, we take a break
to compare two quite different readings of the isomorphism theorem which are available in
our constructive setting. Conventionally, Theorem 3 could be summarised as:

Countable structures with a structure-preserving extension function are isomorphic.

Here the view is that both the input extension function and the output isomorphism
are ordinary (set-theoretic) functions. Moreover, the embedding of the structures into the
natural numbers is a mere limitation to countable cardinality.

The alternative view is that we consider all involved functions to be computable, relying
on the fact that all functions definable in constructive type theory are such.6 This allows for
a synthetic approach to computability [16, 1, 4] disposing of any reference to a formal model
of computation like Turing machines, especially easing the mechanisation of otherwise quite
laborious arguments regarding undecidability [5] and incompleteness [9]. Also in situations
like the back-and-forth method with its set-theoretic reading covering up the inherent
computational content, this approach admits an elegant combination of both perspectives.

So adopting the computational view, the input extension function provides a matching
algorithm µ which we use to implement µ′, Ln, and ultimately F . Since we are able to
implement these functions without classical assumptions,7 their computability is guaranteed.
The only price we have to pay is that the classically trivial case distinction on x ∈ dom(L)
in the definition of µ′ needs to be constructively justified. Fortunately, implementing the
necessary decision procedure for list membership on a type with decidable equality is a
simple exercise, especially if we instead were to hastily use classical assumptions at the
cost of having to program Turing machines for computational results. Finally observing
that a computable embedding into natural numbers replaces countability with effective
enumerability and discreteness (i.e. decidable equality), we may reinterpret Theorem 3:

Computational interpretation: enumerable and discrete structures with a
structure-preserving extension algorithm are computably isomorphic.

6 An intuitive justification for this fact is that constructive type theory is nothing but a dependently
typed programming language. A more formal justification applicable for many constructive systems is
to exhibit a realisability model with the desired properties.

7 So for instance without appeal to the excluded middle (∀P. P ∨ ¬P ), or even stronger, the axiom of
choice. Both principles can be consistently added to constructive type theory but, depending on the
formulation, already the former would allow the definition of uncomputable functions.

https://www.ps.uni-saarland.de/extras/back-and-forth/website/baf.backandforth.html#back_and_forth

This means that in each of the upcoming instantiations we actually obtain two results:
a classical and a computational isomorphism. Moreover, both readings are meaningful in
their own right and it is one of the nice features of constructive mathematics and synthetic
computability that we obtain them both simultaneously after doing the work just once.8

4 Cantor’s Isomorphism Theorem

We now start reaping what we sowed in Section 2 and establish Cantor’s result with our
general isomorphism theorem. To admit the computational view, we represent unbounded
dense linear orders (X, <) with explicit functions d, l, and g for density and unboundedness:

x < d x y < y l x < x < g x

Moreover, we assume a characteristic function f< : X → X → B using the inductive type B
of boolean values such that f< x y = true iff x < y.9 In the computational view, this means
that the witnesses for density and unboundedness can be computed and that the relation is
decidable, while in the classical view using these functions is a mere symbolic convenience.

We now fix two countable unbounded dense linear orders (X, <) and (Y, <) and bring
the informal idea outlined in Section 1 into the shape matching the abstract argument in
Section 2. Given two pairs (x, x′) : X × X and (y, y′) : Y × Y , we define

(x, x′) ∼C (y, y′) := (x = x′ ↔ y = y′) ∧ (x < x′ ↔ y < y′)

and call L : L(X×Y ) a partial order-isomorphism if (x, x′) ∼C (y, y′) for all (x, y), (x′, y′) ∈ L.
By virtue of the abstract isomorphism theorem, we just need to show how a partial order-
isomorphism can be extended by one step in one direction.

▶ Definition 4. We define a function µC : X → L(X × Y ) → Y exactly following the three
cases distinguished in Section 1. So given x ̸∈ dom(L), first use f< to determine which of the
three situations relating x to dom(L) is accurate. In the case where x is in between elements
of dom(L), use d to find a match in Y . In the two other cases where x is below or above all
elements of dom(L), use l and g, respectively. If x ∈ dom(L), we just return a dummy value.

▶ Lemma 5. If x ̸∈ dom(L) and L is a partial order-isomorphism, then so is ((x, µC x L) :: L).

Proof. The only non-trivial obligation is to show (x, x′) ∼C (µC x L, y′) for (x′, y′) ∈ L:
That x = x′ iff µC x L = y′ is straightforward since both are actually false. The former is
ruled out by x ̸∈ dom(L) and the latter since in that case µC x L ̸∈ ran(L).
That x < x′ iff µC x L < y′ follows by analysing the cases underlying the definition of
µC . We only consider the more complicated case where x lies in between elements of
dom(L), the other two cases are similar and easier. In this situation, there are xmax and
xmin in dom(L) with xmax < x < xmin with no domain element in between. Since L is
already a partial order-isomorphism and since then µC x L = d L[xmax] L[xmin], we have
L[xmax] < µC x L < L[xmin] in Y with no range element in between. We then deduce:

x < x′ ⇔ xmin ≤ x′ ⇔ L[xmin] ≤ L[x′] ⇔ µC x L < y′ ◀

Now Cantor’s isomorphism theorem in its classical formulation can be derived directly.

8 As already done in Section 2 we continue to try and use neutral formulations avoiding a bias into one of
the readings, although wordings like “we construct” or “we compute” impose themselves very often.

9 Note that in Coq we prefer to formulate these assumptions with informative types, for instance d is
expressed by ∀xy. Σz. x < z < y and f< could be expressed as ∀xy. (x < y) + (x ̸< y).

ITP 2022

Listing 2 Coq proof script for Cantor’s isomorphism theorem. After using the general isomorphism
theorem back_and_forth, the first eight goals are arranged like the assumptions listed in Listing 1.
The only non-trivial ones are the latter two where partner is µC and step_morph is Lemma 5. The
next five goals instantiate to the given orders and the last goal is the actual derivation of the claim.

Theorem Cantor X Y (OX : dulo X) (OY : dulo Y) (RX : retract X nat) (RY : retract Y nat) :
{ F : X -> Y & { G | inverse F G /\ forall x x', x < x' <-> (F x) < (F x') } }.
Proof.

unshelve edestruct back_and_forth as [F[G[H1 H2]]].
- intros A B. exact (dulo A * dulo B). (* structure *)
- intros A B [OA OB]. exact (OB, OA). (* srev *)
- cbn. intros A B [OA OB]. reflexivity. (* srev_invol *)
- intros A B [OA OB] a a' b b'. exact ((a = a' <-> b = b') /\ (a < a' <-> b < b')). (* iso *)
- cbn. tauto. (* iso_eq *)
- cbn. tauto. (* iso_rev *)
- cbn. intros A B [OA OB] f a. exact (partner OA OB f a). (* find *)
- cbn. intros A B [OA OB] f x. unfold step, tiso. now apply step_morph. (* find_iso *)

- exact X.
- exact Y.
- exact (OX, OY).
- exact RX.
- exact RY.

- cbn in *. exists F, G. split; try apply H1. apply H2.
Qed.

▶ Theorem 6 (Cantor). All countable unbounded dense linear orders are isomorphic.

Proof. We instantiate Theorem 3 as follows, see also Listing 2:
As structure A(X, Y ) we require that X and Y are unbounded dense linear orders.
The inversion operation turning elements of A(X, Y ) to A(Y, X) is obvious.
As isomorphism property we choose (x, x′) ∼C (y, y′).
As polymorphic one-step extension function we choose µC .
Assumptions (1)–(3) are trivial.
Assumption (4) was proven in Lemma 5.

This yields a function F : X → Y with inverse F −1 such that (x, x′) ∼C (F x, F x′). The
latter contains x < x′ ↔ F x < F x′, so F is an order-isomorphism. ◀

The corresponding computational result we obtain for free then reads:

▶ Theorem 7 (Computational Cantor). All decidable linear orders over enumerable domain
with computable witnesses for density and unboundedness are computably isomorphic.

5 Myhill’s Isomorphism Theorem

As a second instance of our general isomorphism theorem, we tackle Myhill’s result that
one-one interreducible sets of natural numbers are recursively isomorphic.10 To first provide
some intuition, recall that a set X ⊆ N many-one reduces to a set Y ⊆ N if there is a
computable function f : N → N such that f(x) ∈ Y implies x ∈ X and f(x) ̸∈ Y implies
x ̸∈ Y . As a typical usage of such a reduction f , a decision procedure for x ∈ X can be
obtained by a decider for Y by first computing f(x) and then determining whether f(x) ∈ Y .
If f happens to be injective (bijective), it is called one-one reduction (recursive isomorphism).

10 We simply adopt the Coq mechanisation given by Forster et al. [3] to our abstract setting and explicitly
link up the corresponding definitions and theorems in the forthcoming text.

https://www.ps.uni-saarland.de/extras/back-and-forth/website/baf.backandforth.html#Cantor
https://www.ps.uni-saarland.de/extras/back-and-forth/website/baf.backandforth.html#Cantor

Given that interreducible sets and recursive isomorphisms are algorithmic notions, we
now fully rely on the synthetic approach to computability as explained in Section 3 and take
simple type-theoretic definitions representations of these notions:

Sets of natural numbers are more generally represented as unary predicates p : X → P on
enumerable and discrete types X (that is, retracts of N).
A many-one reduction from p : X → P to q : Y → P is represented as a function
f : X → Y such that p x iff q (f x), which we denote by f : p ⪯ q.
A one-one reduction (recursive isomorphism) is an injective (bijective) many-one reduction.

Recall that if we were to use classical logic, we would spoil the computational interpretation
and instead would have to explicitly require reductions to be computable in a formal model
like Turing machines. This would turn at least the mechanisation level into a mess, especially
since then the recursive isomorphism obtained by the back-and-forth method must be shown
computable, too.

That said, we now fix two predicates p : X → P and q : Y → P on retracts X and Y

of N, together with one-one reductions f : p ⪯ q and g : q ⪯ p. A recursive isomorphism
F : X → Y will satisfy both F : p ⪯ q and F −1 : q ⪯ p, which yields a strict correspondence
of both predicates. Expressing this correspondence locally for pairs (x, x′) : X × X and
(y, y′) : Y × Y , we therefore want to establish that p x iff q y (by symmetry the same follows
for x′ and y′), along with the usual requirement that the correspondence be one-to-one:

(x, x′) ∼M (y, y′) := (x = x′ ↔ y = y′) ∧ (p x ↔ q y)

We call L : L(X × Y ) a partial (recursive) isomorphism if (x, x′) ∼M (y, y′) for all
(x, y), (x′, y′) ∈ L. Again by virtue of the abstract isomorphism theorem, we just need
to show how a partial isomorphism can be extended by one step in one direction.

▶ Definition 8 (Lemma 55 in [3]). We first define a function µX
M : X → L(X × Y ) → X such

that given x and L it is safe to add f (µX
M x L) as value for x to L. So determine whether

f x ∈ ran(L), if not we can just return x. If otherwise (x′, f x) ∈ L return µX
M x′ L′, where

L′ is L with (x′, f x) removed, so we keep searching until we find some element x0 with
f x0 ̸∈ ran(L) while the search along f ensures that p x iff p x0.11

We then obtain a function µM : X → L(X × Y ) → Y by µM x L := f (µX
M x L).

▶ Lemma 9. If x ̸∈ dom(L) and L is a partial isomorphism, then so is ((x, µM x L) :: L).

Proof. We first verify that µM x = f (µX
M x L) indeed is a suitable match for x:

(p x ↔ p (µX
M x L)) ∧ µX

M x L ∈ x :: dom(L) ∧ f (µX
M x L) ̸∈ ran(L) (∗)

We apply induction following the execution trace of µX
M . In the terminating case where

f x ̸∈ ran(L), we have µX
M x L = x and hence the claim is trivial. In the recursive case we

have µX
M x L = µX

M x′ L′, where L′ is L without (x′, f x), and assume the claim for x′ and L′.
We deduce p x ↔ p (µX

M x′ L′) using p x ↔ q (f x), since f is a reduction, q (f x) ↔ p x′

since L is a partial isomorphism, and p x′ ↔ p (µX
M x′ L′)) by the inductive hypothesis.

We already have µX
M x′ L′ ∈ dom(L) since by induction µX

M x′ L′ ∈ x′ :: dom(L′) and we
have x′ :: dom(L′) = dom(L) by construction of L′.

11 Since this algorithm is not structurally recursive but descends on the length of L, we use the Equations
package [18] to implement it in Coq.

Listing 3 Coq proof script for Myhill’s isomorphism theorem. After using the general isomorphism
theorem back_and_forth, the first eight goals are arranged like the assumptions listed in Listing 1.
The only non-trivial ones are the latter two where mstep is µM and step_corr is Lemma 9. The
next five goals instantiate to the given orders and the last goal is the actual derivation of the claim.

Theorem Myhill X Y (SXY : bireduction X Y) (RX : retract X nat) (RY : retract Y nat) :
{ F : X -> Y & { G | inverse F G /\ reduction p q F } }.
Proof.

unshelve edestruct back_and_forth as [F[G[H1 H2]]].
- intros A B. exact (bireduction A B). (* structure *)
- intros A B S. cbn in *. apply (@Build_bireduction B A eY eX q p g f); apply S. (* srev *)
- cbn. intros A B []. reflexivity. (* srev_invol *)
- intros A B S a a' b b'. cbn in S. exact ((a = a' <-> b = b') /\ (p a <-> q b)). (* iso *)
- cbn. tauto. (* iso_eq *)
- cbn. tauto. (* iso_rev *)
- cbn. intros A B S C a. exact (mstep f eX eY C a). (* find *)
- cbn. intros A B S C a. unfold step, tiso. apply step_corr; apply S. (* find_iso *)

- exact X.
- exact Y.
- exact SXY.
- exact RX.
- exact RY.

- cbn in *. exists F, G. split; try apply H1. intros x. now apply H2.
Qed.

We assume f (µX
M x′ L′) ∈ ran(L), so given that ran(L) = f x :: ran(L′) we have either

f (µX
M x′ L′) ∈ ran(L′) or f (µX

M x′ L′) = f x. Since the former contradicts the inductive
hypothesis the latter must be the case, so µX

M x′ L′ = x by the injectivity of f . Since
the second part of the inductive hypothesis then yields x ∈ x′ :: dom(L′), we obtain a
contradiction to the assumption x ̸∈ dom(L).

Now returning to the actual goal, we show (x, x′) ∼M (µM x L, y′) for (x′, y′) ∈ L.
That x = x′ iff µM x L = y′ is trivial using x ̸∈ dom(L) and µM x L ̸∈ ran(L), the former
being an assumption and the latter part three of (∗).
That p x iff q (µM x L) is immediate using that q (µM x L) iff p (µX

M x L) since f is a
reduction, and p (µX

M x L) iff p x by part one of (∗). ◀

Now Myhill’s isomorphism theorem follows for all enumerable and discrete types.

▶ Theorem 10 (Myhill, Theorem 58 in [3]). One-one interreducible unary predicates on
enumerable and discrete types are recursively isomorphic.

Proof. We instantiate Theorem 3 as follows, see also Listing 3:
As structure A(X, Y ) we require a pair of one-one interreducible predicates.
The inversion operation turning elements of A(X, Y ) to A(Y, X) is obvious.
As isomorphism property we choose (x, x′) ∼M (y, y′).
As polymorphic one-step extension function we choose µM .
Assumptions (1)–(3) are trivial.
Assumption (4) was proven in Lemma 9.

This yields a function F : X → Y with inverse F −1 such that (x, x′) ∼M (F x, F x′). The
latter contains p x ↔ q (F x), so F is a recursive isomorphism. ◀

To conclude, observe that if we forget about the computational view, we have just proven
something quite different: the Cantor-Bernstein theorem restricted to countable sets. This
restriction happens to be provable constructively (exactly due to the connection to Myhill’s
isomorphism theorem), while the general result is equivalent to the excluded middle [14].

https://www.ps.uni-saarland.de/extras/back-and-forth/website/baf.backandforth.html#Myhill

▶ Theorem 11 (Countable Cantor-Bernstein, Theorem 59 in [3]). Countable sets X and Y

with given injections f : X → Y and g : Y → X also admit a bijection h : X → Y .

Proof. If we choose as p and q the empty predicates on X and Y , respectively, then f

trivially satisfies the defining property of a reduction from p to q and analogously so does g

for q and p. Thus Theorem 10 yields a bijection (with a trivial additional property regarding
p and q that we just ignore). ◀

6 Conclusion

In this proof pearl, we have isolated an abstract and computational approach to the back-and-
forth method and considered two prominent instantiations. As the aim was a condensed and
pointed presentation, we refrained from providing more instantiations in other domains for
now. Nevertheless, this brief concluding section is meant to sketch a few more applications
that our approach might be usable for.

First of all, for Cantor’s isomorphism theorem we only discuss the characteristic of dense
linear orders without endpoints, as it is the most familiar case due to its applicability to Q.
However, the three other cases of one (left or right) or two endpoints are completely analogous
and can be obtained by simple modification of our proof. Of course one could also abstract
over the exact characteristic, yielding all four versions simultaneously.

Secondly, regarding other domains, we already mentioned the crucial use of back-and-forth
arguments in model theory, as described in Chapter 7 of [13]. Two further examples are the
isomorphism of countably infinite Boolean algebras [7] as well as the uniqueness of the Rado
graph as the result of a particular method to generate random graphs [15]. We believe that
all these examples can be described in our abstract framework and so for a succinct Coq
mechanisation one would just need to represent the corresponding structure and verify a
suitable step function.

Thirdly, not immediately concerning further instantiations, the computational content of
our axiom-free mechanisation could be made even more explicit by using Coq’s extraction
mechanism. With this feature one would obtain the algorithms implemented in the construct-
ive proofs as executable programs in external programming languages. These algorithms
could then be analysed regarding their efficiency and compared with respect to the necessary
computational primitives. For instance, while the back-and-forth construction itself just
follows the enumeration of the domains, the step function in the case of Cantor’s isomorphism
theorem may exhibit quite different complexity, depending on whether the witnesses for
density and unboundedness are provided as efficient functions or need to be computed with
linear search. For this example and a more general analysis of the back-and-forth method
regarding different notions of computation, we refer the reader to [11].

